

```

/*****
VALLEJO, G., MORIS, J. & CONEJO, N. A SAS/IML PROGRAM FOR ANALYZING
UNIVARIATE AND MULTIVARIATE REPEATED MEASURES DATA. COMPUTER METHODS
AND PROGRAMS IN BIOMEDICINE, 2006.
*****/

```

```

/*-----*/
/* THIS SAS/IML PROGRAM CALCULATES THE F RATIO AND ASSOCIATED P-VALUES FOR OMNIBUS TESTS */
/* AS WELL AS F RATIOS, THE INDIVIDUAL P-VALUES, AND THE HOCHBERG ADJUSTED P-VALUES FOR */
/* PAIRWISE COMPARISONS OF THE MARGINAL MEANS AND MULTIPLE POSSIBLE PRODUCT INTERACTION */
/* CONTRASTS, FOR A VARIETY OF UNIVARIATE AND MULTIVARIATE REPEATED MEASURES DESIGNS. ALL */
/* OF THE F TESTS AND THE HOCHBERG ADJUSTED P-VALUES ARE CALCULATE AUTOMATICALLY BY THE */
/* PROGRAM.
/*-----*/

```

```

/*-----*/
/*-----REQUIRED ARGUMENTS ON FIRST USE-----*/
/*-----*/
/* NDV = NUMBER OF DEPENDENT VARIABLES
/*-----*/

```

```

/*-----*/
/*-----OPTIONAL ARGUMENTS-----*/
/*-----*/
/* TESTOMNI=1, WHERE VALUE IS EITHER 1 OR 0. IF TESTOMNI=1, THEN THE F TESTS FOR THE MAIN */
/* AND INTERACTION EFFECTS ARE COMPUTED. DEFAULT=1.
/* TESTGROUPS=1, WHERE VALUE IS EITHER 1 OR 0. IF TESTGROUPS=1, THEN ALL PAIRWISE CONTRASTS */
/* AMONG LEVELS OF THE GROUPS ARE COMPUTED. DEFAULT=1.
/* TESTTIME=1, WHERE VALUE IS EITHER 1 OR 0. IF TESTTIME = 1, THEN ALL PAIRWISE CONTRASTS */
/* AMONG LEVELS OF THE TRIAL ARE COMPUTED. DEFAULT=1.
/* TESTINTERAC =1, WHERE VALUE IS EITHER 1 OR 0. IF TESTINTERAC=1, THEN MULTIPLE POSSIBLE */
/* PRODUCT INTERACTION CONTRASTS ARE COMPUTED. DEFAULT = 1.
/* TESTDEPVAR=1, WHERE VALUE IS EITHER 1 OR 0. IF TESTDEPVAR=1, THEN MULTIVARIATE OMNIBUS */
/* TESTS WITH MULTIPLE POST HOC CONTRASTS ARE COMPUTED. ALSO ARE COMPUTED OMNIBUS TESTS */
/* FOR EACH DEPENDENT OUTCOME AS WELL AS MULTIPLE COMPARISON HYPOTHESES RELATED TO THESE */
/* EFFECTS. DEFAULT =1.
/*-----*/

```

```
DATA DATARECORDED; INPUT GROUP Y1-Y5; CARDS;
```

```

/*****
/** CHANGE THIS VARIABLES FOR THE DATA BEING USED **/
*****/

```

```
PROC IML;
```

```
USE DATARECORDED;
```

```

NDV=1;          /*NUMBER OF DEPENDANT VARIABLES*/
TESTOMNI=1;    /*1 OMNIBUS TESTS, 0 NO OMNIBUS TESTS*/
TESTGROUPS=1; /*1 GROUPS PAIRWISE CONTRASTS, 0 NO GROUPS PAIRWISE CONTRASTS*/
TESTTIME=1;   /*1 TIME PAIRWISE CONTRASTS, 0 NO TIME PAIRWISE CONTRASTS*/
TESTINTERAC=1; /*1 TESTING INTERACTION CONTRASTS, 0 TESTING INTERACTION CONTRASTS*/
TESTDEPVAR=1; /*1 UNIVARIATE TEST OF EACH OF THE NDV, 0 NO TEST*/

```

```
START MBF (GROUP1, GROUP2, TIME1, TIME2, RESULTSV, NGROUPS, NOBS, TESTDV) GLOBAL (NNDV);
```

```

/*****
** INITIAL CALCULATIONS **
*****/
TNDV=NNDV;
IF ((GROUP1=-1)&(GROUP2=-1)) THEN READ ALL INTO TEMPY;      /*TESTING PAIRWISE
                                                                GROUPS/INTERACTION */
    ELSE DO;
        READ ALL where ((GROUP=GROUP1)|(GROUP=GROUP2))INTO TEMPY;
        DO I=1 TO (NROW(TEMPY));
            IF TEMPY [I,1] = GROUP1 THEN TEMPY[I,1]=1;
            IF TEMPY [I,1] = GROUP2 THEN TEMPY[I,1]=2;
        END;
    END;

COLUMNS=NCOL(TEMPY);
ROWS=NROW(TEMPY);

IF ((TIME1>-1)&(TIME2>-1)) THEN DO; /*PAIRWISE OBSERVATIONS/INTERACTION*/
    FLEP=TEMPY[,1];
    DO I=1 TO TNDV;
        FLEP=FLEP||TEMPY[,TIME1+1+((I-1)*((NCOL(TEMPY)-
        1)/TNDV))||TEMPY[,TIME2+1+((I-1)*((NCOL(TEMPY)-1)/TNDV))];
    END;
    TEMPY=FLEP;
END;

IF (TESTDV) THEN DO;
    FLEP=TEMPY[,1];
    DO I=1 TO ((NCOL(TEMPY))-1)/TNDV;
        FLEP=FLEP||TEMPY[,I+1+((TESTDV-1)*((NCOL(TEMPY)-1)/TNDV))];
    END;
    TEMPY=FLEP;
    TNDV=1;
END;

NG=NCOL(UNION(TEMPY[,1]));          /* NUMBER OF GROUPS */
NGROUPS=NG;
DF=J(NG,1,0);
DO I=1 TO NG;
    NOBS=0;
    DO J=1 TO ROWS;
        IF TEMPY [J,1] = I THEN DO; /* NUMBER OF SUBJECTS IN GROUP I*/
            NOBS = NOBS + 1;
        END;
    END;
    DF(|I|)=NOBS-1;
END;
DFF=DF+1;                          /* MATRIX RELATED TO FREEDOM DEEGRES*/

Y=TEMPY[,2];
IF NCOL(TEMPY)> 2 THEN DO;
    DO TEMP=3 TO NCOL(TEMPY);
        Y=Y||TEMPY[,TEMP];          /*THIS LOOP READS THE DATA RECORDED INTO...*/
    END;                            /*...THE MATRIX Y*/
END;
NT=NCOL(Y)/TNDV;                    /* NUMBER OF OBSERVATIONS (TIME) */
NOBS=NT;
NTOT=NROW(Y);                       /* TOTAL NUMBER SUBJECTS */

```

```

DO I=1 TO NG;
  TEMP=J(DFF[I],1,I);
  IF I=1 THEN DO;
    XX=TEMP;
  END;
  ELSE DO;
    XX=XX//TEMP;
  END;
END;

X = DESIGN(XX); /* DESIGN MATRIX*/
CG = J(NG-1,1,1) || (-I(NG-1)); /* GROUP EFFECT CONTRAST MATRIX*/
CT = J(1,NG,1); /* TIME CONTRAST MATRIX */
CGT = J(NG-1,1,1) || (-I(NG-1)); /* INTERACTION CONTRAST MATRIX*/
A=(I(TNDV))@(J(NT-1,1,1) || (-I(NT-1))); /* POST MATRIX FOR TIME CONTRAST*/
U = (I(TNDV))@J(1,NT,1); /*POST MATRIX FOR GROUPS CONTRAST*/

/*****
** COMPUTE THE ORDINARY LEAST SQUARES AND MODEL SUM OF SQUARES **
*****/

B = INV(X`*X)*X`*Y;
D = INV(DIAG(DFF));
BT = SQRT(NTOT*(SUM(1/(DFF))))*B;
TJ = 1/(DFF);
CJ = 1-(DFF)/NTOT;
RJ = (DFF)/NTOT;
ZJ = NTOT/(DFF);

VQG=0;VQT=0;VQGT=0;

DO I=1 TO NG;
  RG=U*((T(Y#X[,I]-X[,I]*B[I,]))*(Y#X[,I]-X[,I]*B[I,]))/(DF[I])*U`;
  RI=A*((T(Y#X[,I]-X[,I]*B[I,]))*(Y#X[,I]-X[,I]*B[I,]))/(DF[I])*A`;
  VQG=VQG+(CJ[I]*RG);
  VQT=VQT+(TJ[I]*RI);
  VQGT=VQGT+(CJ[I]*RI);
END;

HG = T(CG*B*U`) * INV(CG*D*CG`) * (CG*B*U`); /*SC GROUP*/
HT = T(CT*BT*A`) * INV(CT*D*CT`) * (CT*BT*A`); /*SCPC TIME*/
HGT= T(CGT*B*A`) * INV(CGT*D*CGT`) * (CGT*B*A`); /*SCPC INTERACTION*/
VG=INV(VQG);
VT=INV(VQT);
VGT=INV(VQGT);

/*****
** COMPUTE DISPERSION MATRICES AND ASSOCIATED DEGREES OF FREEDOM **
*****/

/*INITIALIZATION OF MATRICES USED IN THE CALCULATION*/
VHG2=0;VHG7=0;VHGT7=0;VHGT2=0;VEEG=0;VEET=0;VEEGT=0;EEG=0;EET=0;EEGT=0;

/*CALCULATION OF THE ERROR MATRICES AND THE DF*/
DO I=1 TO NG;
  SG=U*((T(Y#X[,I]-X[,I]*B[I,]))*(Y#X[,I]-X[,I]*B[I,]))/(DF[I])*T(U);

```

```

VGE=CJ[I]* (SG*VG);
VEEG=VEEG+VGE;
VEG1= ( (TRACE (VEEG*VEEG)) + (TRACE (VEEG)##2) );
VGEE= ( (TRACE (VGE*VGE)) + (TRACE (VGE)##2) ) / (DF (|I|));
VEG2=SUM (VEG2//VGEE);
VHG1=RJ[I]* (SG*VG);
VHG2=VHG2+VHG1;
VHG3= ( (TRACE (VHG2*VHG2)) + (TRACE (VHG2)##2) );
VHG4= (SG*VG);
VHG5= ( (TRACE (VHG4*VHG4)) + (TRACE (VHG4)##2) );
VHG6=SUM (VHG6//VHG5);
VHG7=VHG7+2*(VHG5*RJ[I]);
SI=A* ( (T(Y#X[,I]-X[,I]*B[I,])*(Y#X[,I]-X[,I]*B[I,])) / (DF[I]))*T(A);

VTT=TJ[I]* (SI*VT);
VEET=VEET+VTT;
VET1= ( (TRACE (VEET*VEET)) + (TRACE (VEET)##2) );
VET2= ( (TRACE (VTT*VTT)) + (TRACE (VTT)##2) ) / (DF (|I|));
VET3=SUM (VET3//VET2);

VGTT=CJ[I]* (SI*VGT);
VEEGT=VEEGT+VGTT;
VEGT0= ( (TRACE (VEEGT*VEEGT)) + (TRACE (VEEGT)##2) );
VEGT1= ( (TRACE (VGTT*VGTT)) + (TRACE (VGTT)##2) ) / (DF (|I|));
VEGT2=SUM (VEGT2//VEGT1);
VHGT1=RJ[I]* (SI*VGT);
VHGT2=VHGT2+VHGT1;
VHGT3= ( (TRACE (VHGT2*VHGT2)) + (TRACE (VHGT2)##2) );
VHGT4= (SI*VGT);
VHGT5= ( (TRACE (VHGT4*VHGT4)) + (TRACE (VHGT4)##2) );
VHGT6=SUM (VHGT6//VHGT5);
VHGT7=VHGT7+2*(VHGT5*RJ[I]);
EEG1=CJ[I]* (SG);
EEG=EEG+EEG1;
EET1=ZJ[I]* (SI);
EET=EET+EET1;
EEGT1=CJ[I]* (SI);
EEGT=EEGT+EEGT1;

END;

VEG=VEG1/VEG2; /*ERROR DF FOR BETWEEN GROUPS*/
VHG=VEG1 / (VHG3+VHG6 - VHG7); /*HIPOTHESIS DF FOR BETWEEN GROUPS*/
VET=VET1/VET3; /*ERROR DF FOR WITHIN SUBJECTS*/
VEGT=VEGT0/VEGT2; /*ERROR DF FOR INTERACTION*/
VHGT=VEGT0 / (VHGT3+VHGT6 - VHGT7); /*HIPOTHESIS DF FOR INTERACTION*/

EG=(VEG/VHG)*(EEG); /*ERROR MATRIX BETWEEN GROUPS */
ET=(VET/1)*(EET); /*ERROR MATRIX WITHIN SUBJECTS */
EGT=(VEGT/VHGT)*(EEGT); /*ERROR MATRIX INTERACTION */

/*****
** COMPUTE DEGREES OF FREEDOM, F-STATISTICS, AND P-VALUES **
*****/

/*BETWEEN-GROUPS DF, F AND P*/
MG=ROUND (TRACE (GINV (EG) *EG)); SG=1;
IF (VHG>1) THEN DO;
SG= ( (MG##2*VHG##2-4) / (MG##2+VHG##2-5) )##.5;END;IF SG=. THEN DO; SG=1;END;

```

```

MG=ROUND(TRACE(GINV(EG)*EG));
DFG1=MG*VHG; /*DF1*/
DFG2=(VEG-(MG-VHG+1)/2)*SG-(MG*VHG-2)/2; /*DF2*/
LMDG=(DET(EG)/DET(HG+EG));
FVALG=((1-LMDG)/LMDG)*(DFG2/DFG1); /*F VALUE*/
PVALG=1-PROBF(FVALG,DFG1,DFG2); /*P VALUE*/

/*WITHIN-SUBJECTS DF, F AND P*/
MT=ROUND(TRACE(GINV(CT)*CT));
DFT1=ROUND(TRACE(GINV(ET)*ET)); /*DF1*/
DFT2=(VET-(DFT1-MT+1)/2-(DFT1*MT-2)/2); /*DF2*/
LMDT=(DET(ET)/DET(HT+ET));
FVALT=((1-LMDT)/LMDT)*(DFT2/DFT1); /*F VALUE*/
PVALT=1-PROBF(FVALT,DFT1,DFT2); /*P VALUE*/

/*INTERACTION DF, F AND P*/
MGT=ROUND(TRACE(GINV(EGT)*EGT));SGT=1;
IF (VHGT>1) THEN DO;
SGT=((MGT##2*VHGT##2-4)/(MGT##2+VHGT##2-5))##.5;END;IF SGT=. THEN DO; SGT=1;END;
DGT1=MGT*VHGT; /*DF1*/
DGT2=(VEGT-(MGT-VHGT+1)/2)*SGT-(MGT*VHGT-2)/2; /*DF2*/
LMDGT=(DET(EGT)/DET(HGT+EGT));
FVAGT=((1-LMDGT##(1/SGT))/LMDGT##(1/SGT))*(DGT2/DGT1); /*F VALUE*/
PVAGT=1-PROBF(FVAGT,DGT1,DGT2); /*P VALUE*/

/*RESULTS VECTOR*/
RESULTSV=REPEAT(0,3,4);
RESULTSV[1,1]=DFG1; RESULTSV[1,2]=DFG2; RESULTSV[1,3]=FVALG;
RESULTSV[1,4]=PVALG;
RESULTSV[2,1]=DFT1; RESULTSV[2,2]=DFT2; RESULTSV[2,3]=FVALT;
RESULTSV[2,4]=PVALT;
RESULTSV[3,1]=DGT1; RESULTSV[3,2]=DGT2; RESULTSV[3,3]=FVAGT; RESULTSV[3,4]=PVAGT;

FINISH;

/*****
** PRESENTATION OF THE RESULTS **
*****/

START OMNIRERESULTS (RESULTSV,TESTDV);

DFG1=RESULTSV[1,1]; DFG2=RESULTSV[1,2]; FVALG=RESULTSV[1,3];
PVALG=RESULTSV[1,4];
DFT1=RESULTSV[2,1]; DFT2=RESULTSV[2,2]; FVALT=RESULTSV[2,3];
PVALT=RESULTSV[2,4];
DGT1=RESULTSV[3,1]; DGT2=RESULTSV[3,2]; FVAGT=RESULTSV[3,3];
PVAGT=RESULTSV[3,4];

RESET NONAME; RESET CENTER; PRINT '';
PRINT 'MODIFIED BROWN-FORSYTHE OMNIBUS TEST RESULTS';
IF (TESTDV) THEN DO; PRINT ''; PRINT 'UNIVARIATE RESULTS FOR DEPENDANT VARIABLE '
TESTDV; END;
RESET NOCENTER;
PRINT '
VALUE' ;
DF F-VALUE P-
PRINT "GROUPS " (' DFG1 ', ' DFG2 ') | ' FVALG ' | ' PVALG;
PRINT "TRIALS " (' DFT1 ', ' DFT2 ') | ' FVALT ' | ' PVALT;
PRINT "GROUPS*TRIALS " (' DGT1 ', ' DGT2 ') | ' FVAGT ' | ' PVAGT;

```

```

FINISH;

START GROUPTTEST (P, NGROUPS,TESTDV);
  IF ((P<(0.05))&(NGROUPS>2)) THEN DO;
    RESULTSV=REPEAT(0,3,4);
    COMBG=((NGROUPS*(NGROUPS-1))/2);
    GROUPV=REPEAT(0,COMBG,6);
    DO I=1 TO (NGROUPS-1);
      DO J=I+1 TO NGROUPS;
        GROUPV[COMBG,1]=I;
        GROUPV[COMBG,2]=J;
        RUN MBF (I, J, -1, -1, RESULTSV, 0, 0, TESTDV);
        GROUPV[COMBG,3]=RESULTSV[1,1];
        GROUPV[COMBG,4]=RESULTSV[1,2];
        GROUPV[COMBG,5]=RESULTSV[1,3];
        GROUPV[COMBG,6]=RESULTSV[1,4];
        COMBG=COMBG-1;
      END;
    END;
  END;
DO I=1 TO (NROW(GROUPV)-1);
  DO J=1 TO (NROW(GROUPV)-I);
    IF ((GROUPV[J,6])>(GROUPV[J+1,6])) THEN DO;
      TEMP=GROUPV[J,]; GROUPV[J,]=GROUPV[J+1,]; GROUPV[J+1,]=TEMP;
    END;
  END;
END;
IF ((P<(0.05))&(NGROUPS>2)) THEN DO;
  PRINT ''; PRINT 'GROUPS PAIRWISE COMPARISONS WITH HOCHBERG''''S H-ADJ-
VALUES';
  IF (TESTDV) THEN DO; PRINT 'UNIVARIATE RESULTS FOR DEPENDANT VARIABLE '
TESTDV; END;
  TEMP=0;
  DO I=1 TO (NROW(GROUPV));
    GROUP1=GROUPV[I,1];GROUP2=GROUPV[I,2];
    DF1=GROUPV[I,3];DF2=GROUPV[I,4];
    FVALUE=GROUPV[I,5];PVALUE=GROUPV[I,6];
    ADJUSTEDP=GROUPV[I,6]*(NROW(GROUPV)+1-I);
    IF (ADJUSTEDP>0.05) THEN TEMP=1;
    IF (TEMP) THEN ADJUSTEDP=GROUPV[NROW(GROUPV),6];
    IF I=1 THEN PRINT GROUP1 [format=2.0] '' GROUP2 [format=2.0]
[LABEL='TEST'] '' DF1 [LABEL='DF1'] [format=5.3] ''
DF2 [format=5.3] [LABEL='DF2'] '' FVALUE [format=5.4] [LABEL='F
VALUE'] '' PVALUE [LABEL='P VALUE'] '' ADJUSTEDP [LABEL='H-ADJ'];
    ELSE PRINT GROUP1 [format=2.0] '' GROUP2 [format=4.0]
'' DF1 [format=5.3] '' DF2 [format=5.3] '' FVALUE [format=5.4] ''
'' PVALUE '' ADJUSTEDP;
  END;
END;
FINISH;

START TIMETEST (P, NOBS, TESTDV);
  IF ((P<(0.05))&(NOBS>2)) THEN DO;
    RESULTSV=REPEAT(0,3,4);
    COMBT=((NOBS*(NOBS-1))/2);
    TIMEV=REPEAT(0,COMBT,6);
    DO I=1 TO (NOBS-1);

```

```

DO J=I+1 TO NOBS;
TIMEV[COMBT,1]=I;
TIMEV[COMBT,2]=J;
RUN MBF (-1, -1, I, J, RESULTSV, 0, 0, TESTDV);
TIMEV[COMBT,3]=RESULTSV[2,1];
TIMEV[COMBT,4]=RESULTSV[2,2];
TIMEV[COMBT,5]=RESULTSV[2,3];
TIMEV[COMBT,6]=RESULTSV[2,4];
COMBT=COMBT-1;
END;
END;
END;
DO I=1 TO (NROW(TIMEV)-1);
DO J=1 TO (NROW(TIMEV)-I);
IF ((TIMEV[J,6])>(TIMEV[J+1,6])) THEN DO;
TEMP=TIMEV[J,]; TIMEV[J,]=TIMEV[J+1,]; TIMEV[J+1,]=TEMP;
END;
END;
END;
IF ((P<(0.05))&(NOBS>2)) THEN DO;
PRINT ''; PRINT 'OBSERVATIONS PAIRWISE COMPARISONS WITH HOCHBERG''S H-ADJ-
VALUES';
IF (TESTDV) THEN DO; PRINT 'UNIVARIATE RESULTS FOR DEPENDANT VARIABLE '
TESTDV; END;
TEMP=0;
DO I=1 TO (NROW(TIMEV));
TIME1=TIMEV[I,1];TIME2=TIMEV[I,2];
DF1=TIMEV[I,3];DF2=TIMEV[I,4];
FVALUE=TIMEV[I,5];PVALUE=TIMEV[I,6];
ADJUSTEDP=TIMEV[I,6]*(NROW(TIMEV)+1-I);
IF (ADJUSTEDP>0.05) THEN TEMP=1;
IF (TEMP) THEN ADJUSTEDP=TIMEV[NROW(TIMEV),6];
IF I=1 THEN PRINT TIME1 [format=2.0] '' TIME2 [format=2.0]
[LABEL='TEST'] '' DF1 [LABEL='DF1'] [format=5.3] ''
DF2 [format=5.3] [LABEL='DF2'] '' FVALUE [format=5.4] [LABEL='F
VALUE'] '' PVALUE [LABEL='P VALUE'] '' ADJUSTEDP [LABEL='H-ADJ'];
ELSE PRINT TIME1 [format=2.0] '' TIME2 [format=4.0] '' DF1
[format=5.3] ''
DF2 [format=5.3] '' FVALUE [format=5.4] '' PVALUE '' ADJUSTEDP;
END;
END;
FINISH;
START INTERACTEST (P, NGROUPS, NOBS,TESTDV);
IF ((P<(0.05))&((NGROUPS+NOBS)>4)) THEN DO;
RESULTSV=REPEAT(0,3,4);
COMBI=((NGROUPS*(NGROUPS-1))/2)*((NOBS*(NOBS-1))/2);
INTERACV=REPEAT(0,COMBI,8);
DO I=1 TO (NGROUPS-1);
DO J=I+1 TO NGROUPS;
DO K=1 TO (NOBS-1);
DO L=K+1 TO NOBS;
INTERACV[COMBI,1]=I;
INTERACV[COMBI,2]=J;
INTERACV[COMBI,3]=K;
INTERACV[COMBI,4]=L;
RUN MBF (I, J, K, L, RESULTSV, 0, 0, TESTDV);

```

```

INTERACV[COMBI,5]=RESULTSV[3,1];
INTERACV[COMBI,6]=RESULTSV[3,2];
INTERACV[COMBI,7]=RESULTSV[3,3];
INTERACV[COMBI,8]=RESULTSV[3,4];
COMBI=COMBI-1;
END;
END;
END;
END;
DO I=1 TO (NROW(INTERACV)-1);
DO J=1 TO (NROW(INTERACV)-I);
IF ((INTERACV[J,8])>(INTERACV[J+1,8])) THEN DO;
TEMP=INTERACV[J,]; INTERACV[J,]=INTERACV[J+1,];
INTERACV[J+1,]=TEMP;
END;
END;
END;
END;
DO I=1 TO (NROW(INTERACV));
GROUP1=INTERACV[I,1];GROUP2=INTERACV[I,2];
TIME1=INTERACV[I,3]; TIME2=INTERACV[I,4];
DF1=INTERACV[I,5];DF2=INTERACV[I,6];
FVALUE=INTERACV[I,7];PVALUE=INTERACV[I,8];
ADJUSTEDP=INTERACV[I,8]*(NROW(INTERACV)+1-I);
IF ((ADJUSTEDP>0.05)&(TEMP=0)) THEN TEMP=1; IF (TEMP) THEN
ADJUSTEDP=INTERACV[NROW(INTERACV),8];
IF I=1 THEN PRINT GROUP1 [format=2.0] ' ' GROUP2 [format=2.0]
[LABEL='GROUPS'] ' ' TIME1 [format=2.0] ' ' TIME2 [format=2.0] [LABEL='OBS.'] ' ' DF1
[format=5.3] [LABEL='DF1']
' ' DF2 [format=5.3][LABEL='DF2'] ' ' FVALUE [format=5.3] [LABEL='F
VALUE'] ' ' PVALUE [LABEL='P VALUE'] ' ' ADJUSTEDP [LABEL='H-ADJ'];
ELSE PRINT GROUP1 [format=2.0] ' ' GROUP2 [format=6.0] ' ' TIME1
[format=2.0] ' ' TIME2 [format=4.0] ' ' DF1 [format=5.3]
' ' DF2 [format=5.3] ' ' FVALUE [format=5.3] ' ' PVALUE ' '
ADJUSTEDP;
END;
END;
FINISH;

START DEPVARTEST (TESTDV) GLOBAL (NDV);
IF (NDV>1) THEN DO;
DO TESTDV=1 TO NDV;
RUN MBF (-1, -1, -1, -1, RESULTSV, TNGROUPS, TNOBS, TESTDV);
RUN OMNIRERESULTS (RESULTSV, TESTDV);
P=RESULTSV[1,4]; RUN GROUPTTEST (P,TNGROUPS,TESTDV);
P=RESULTSV[2,4]; RUN TIMETEST (P,TNOBS,TESTDV);
P=RESULTSV[3,4]; RUN INTERACTEST (P,TNGROUPS,TNOBS,TESTDV);
END;
END;
FINISH;

```

```

/*****
** MAIN **
*****/

/*VARIABLE INITIALIZATION*/
GROUP1=-1; GROUP2=-1; TIME1=-1; TIME2=-1; P=0; TESTDV=0;
NGROUPS=0; NOBS=0; GROUPV=0; TIMEV=0; RESULTSV=0;NNDV=NDV;

RUN MBF (GROUP1, GROUP2, TIME1, TIME2, RESULTSV, NGROUPS, NOBS, 0); /*OMNIBUS TESTS*/
IF (TESTOMNI) THEN RUN OMNIRERESULTS (RESULTSV, TESTDV); /*OMNIBUS RESULTS*/
P=RESULTSV[1,4];
IF (TESTGROUPS) THEN RUN GROUPTEST (P,NGROUPS,TESTDV); /*BETWEEN-SUBJECTS CONTRASTS*/
P=RESULTSV[2,4];
IF (TESTTIME) THEN RUN TIMETEST (P,NOBS,TESTDV); /*WITHIN-SUBJECTS CONTRASTS*/
P=RESULTSV[3,4];
IF (TESTINTERAC) THEN RUN INTERACTEST (P,NGROUPS,NOBS,TESTDV);/*INTERACTION CONTRASTS*/
IF (TESTDEPVAR) THEN RUN DEPVARTEST (TESTDV); /*DEPENDANT VARIABLES...*/
/*...UNIVARIATE TEST*/

PROC PRINT;

```